>

Last time, saw that classification of a linear system like

> $\langle D(x), D(y)\rangle = A.\langle x, y\rangle$

$$\begin{bmatrix} D(x) \\ D(y) \end{bmatrix} = A.\begin{bmatrix} x \\ y \end{bmatrix} \tag{1}$$

is determined by eigenvalues (and eigenvectors) of A: λ_1, λ_2

$\lambda_1, \lambda_2 > 0$, source

$\lambda_1, \lambda_2 < 0$, sink

$\lambda_1 > 0,\ \lambda_2 < 0$ *saddle*

$\lambda_1 = \lambda_2$ *degenerate,* $\cdots$ *higher order terms can* **break** *it.*

λ_1, λ_2 complex $\lambda_1 = a + bi, \lambda_2 = a - bi$

$a > 0$, *spiral source*
a < 0 spiral sink
a = 0 (elliptic) ***** if linear. Higher order terms can break this.

Trace = sum of diagonal elements = $\lambda_1 + \lambda_2$

Determinant = $\lambda_1 \cdot \lambda_2$

Eigenvalues of A $= \dfrac{trace \pm \sqrt{(trace)^2 - 4\,det}}{2}$

>

>

--------------------- What is below has a problem, and is here for historical reasons ----------------
I'll put it in gray and a smaller font so you can see where the wrong stuff ends easily.

```
> phug:=(theta,v)->[ (v^2-cos(theta))/v, -sin(theta)-R*v^2];
```

$$phug := (\theta, v) \to \left[\frac{v^2 - \cos(\theta)}{v}, -\sin(\theta) - R\,v^2\right] \tag{2}$$

```
> Jack:=VectorCalculus[Jacobian](phug(theta,v), [theta,v]);
```

$$Jack := \begin{bmatrix} \frac{\sin(\theta)}{v} & 2 - \frac{v^2 - \cos(\theta)}{v^2} \\ -\cos(\theta) & -2\,R\,v \end{bmatrix} \tag{3}$$

```
> convert(
    solve( {phug(theta,v)[1]=0, phug(theta,v)[2]=0}, [theta,v]), radical);
```

$$\left[\left[\theta = \arctan\left(-\sqrt{\frac{1}{R^2+1}}\,R, \sqrt{\frac{1}{R^2+1}}\right), v = \left(\frac{1}{R^2+1}\right)^{1/4}\right]\right] \tag{4}$$

```
> Fix:=R->[arctan(-R*sqrt(1/(R^2+1))),(1/(R^2+1))^(1/4)];
```

$$Fix := R \to \left[\arctan\left(-\sqrt{\frac{1}{R^2+1}}\, R \right), \left(\frac{1}{R^2+1} \right)^{1/4} \right] \quad \textbf{(5)}$$

```
> FixMat:=unapply(eval(Jack, {theta=Fix(R)[1], v=Fix(R)[2]}),R):
> with(LinearAlgebra):
```

```
> Trace(Jack);
```

$$\frac{\sin(\theta)}{v} - 2\,R\,v \quad \textbf{(6)}$$

```
> subs({theta=Fix(R)[1], v=Fix(R)[2]},%);
```

$$\frac{\sin\left(-\arctan\left(\sqrt{\frac{1}{R^2+1}}\, R \right) \right)}{\left(\frac{1}{R^2+1} \right)^{1/4}} - 2\,R\left(\frac{1}{R^2+1} \right)^{1/4} \quad \textbf{(7)}$$

```
> simplify(%,trig);
```

$$-\frac{\left(\frac{1}{R^2+1} \right)^{1/4} R}{\sqrt{1+\frac{R^2}{R^2+1}}} - 2\,R\left(\frac{1}{R^2+1} \right)^{1/4} \quad \textbf{(8)}$$

Somethings wrong. I'll fi8x this later. life sucks.

---------------- let's try this again (added on April 16)

What I want to do is determine how the type of fixed point varies with the parameter R.

First, we enter our system, and solve for the v and theta which correspond to the fixed point, as a function of R. This was done correctly above, but I'll redo it here for continuity.

```
> phug:=(theta,v)->[v-cos(theta)/v, -sin(theta)-R*v^2];
  Jack:=VectorCalculus[Jacobian](phug(theta,v),[theta,v]);
```

$$phug := (\theta, v) \to \left[v - \frac{\cos(\theta)}{v},\ -\sin(\theta) - R\,v^2 \right]$$

$$Jack := \begin{bmatrix} \frac{\sin(\theta)}{v} & 1 + \frac{\cos(\theta)}{v^2} \\ -\cos(\theta) & -2\,R\,v \end{bmatrix} \quad \textbf{(9)}$$

What I did wrong above (which was stupid) was to rewrite the solution in terms of a version of arctan which ignored the second argument (which identifies the branch). This means that maple is unable to simplify the expression cleanly. So let's try it in a somewhat different way:

```
> with(LinearAlgebra):
> FixSols:= convert(solve(phug(theta,v),{theta,v}), radical);
```

$$FixSols := \left\{ v = \left(\frac{1}{R^2+1} \right)^{1/4},\ \theta = \arctan\left(-\sqrt{\frac{1}{R^2+1}}\, R,\ \sqrt{\frac{1}{R^2+1}} \right) \right\} \quad \textbf{(10)}$$

```
> simplify(subs(FixSols,Jack));
  JacFix:=unapply(%,R):
```

$$\left[\begin{array}{cc} -R\left(\frac{1}{R^2+1}\right)^{1/4} & 2 \\ -\sqrt{\frac{1}{R^2+1}} & -2\,R\left(\frac{1}{R^2+1}\right)^{1/4} \end{array}\right] \tag{11}$$

```
> JacFix(0);Eigenvalues(%);
```

$$\left[\begin{array}{cc} 0 & 2 \\ -1 & 0 \end{array}\right]$$

$$\left[\begin{array}{c} \mathrm{I}\sqrt{2} \\ -\mathrm{I}\sqrt{2} \end{array}\right] \tag{12}$$

```
> JacFix(3.0); Eigenvalues(JacFix(3.0));
```

$$\left[\begin{array}{cc} -1.687023976 & 2 \\ -0.3162277660 & -3.374047952 \end{array}\right]$$

$$\left[\begin{array}{c} -2.24936530070151+0.\,\mathrm{I} \\ -2.81170662729849+0.\,\mathrm{I} \end{array}\right] \tag{13}$$

That's more like it!

```
> EV:=simplify(Eigenvalues(JacFix(R)));
```

$$EV := \left[\begin{array}{c} -\frac{3}{2}\,R\left(\frac{1}{R^2+1}\right)^{1/4}+\frac{1}{2}\sqrt{\sqrt{\frac{1}{R^2+1}}\,\left(-8+R^2\right)} \\ -\frac{3}{2}\,R\left(\frac{1}{R^2+1}\right)^{1/4}-\frac{1}{2}\sqrt{\sqrt{\frac{1}{R^2+1}}\,\left(-8+R^2\right)} \end{array}\right] \tag{14}$$

Here we can see that we will have complex eigenvalues whenever $R^2 < 8$, that is, for $|R| < 2\sqrt{2}$. In addition, for $0 < R < 2\sqrt{2}$, the eigenvalues will have a negative real part, which means we have a spiral sink (solutions spiral in to the fixed point). For $R > 2\sqrt{2}$, both eigenvalues are real. While it is obvious that the second eigenvalue is always negative, it isn't completely obvious that the first is. We can get a clue by making a plot:

```
> plot(EV,R=2*sqrt(2)..9, y=-6..0);
```



Now let's think a little to confirm our suspicions. Note that the factor $\frac{1}{2}\left(\frac{1}{R^2+1}\right)^{1/4}$ can be pulled out of both terms of EV[1] (even though maple is reluctant to do so), which means that the eigenvalues are

$$\frac{1}{2}\left(\frac{1}{R^2+1}\right)^{1/4}\left(-3\,R \pm \sqrt{R^2-8}\,\right)$$

Since $3\,R > \sqrt{R^2-8}$ for all values of $R > 2\sqrt{2}$, we know both eigenvalues are always negative.

This means for $R > 0$, the fixed point is always a sink (which is spiral for $0 < R < 2\sqrt{2}$)

---------------------------- now back to what we did in class. --------------------------------------

```
> with(plots):
  with(DEtools):
> xphug:=R-> [diff(theta(t),t) = (v(t)^2 - cos(theta(t)))/(v(t)),
              diff(v(t),t)      = -sin(theta(t))-R*v(t)^2,
              diff(x(t),t)      = v(t)*cos(theta(t)),
              diff(y(t),t)      = v(t)*sin(theta(t))];
```

$$xphug := R \rightarrow \left[\frac{d}{dt}\,\theta(t) = \frac{v(t)^2 - \cos(\theta(t))}{v(t)},\ \frac{d}{dt}\,v(t) = -\sin(\theta(t)) - R\,v(t)^2,\ \frac{d}{dt}\,x(t) = v(t)\,\cos(\theta(t)),\ \frac{d}{dt}\,y(t) = v(t)\,\sin(\theta(t))\right] \quad \textbf{(15)}$$

```
> Plot4 := proc(vmin, vmax, numv, {yrange:=0..4, R:=0, xrange:=-1..4})
   local inits,vi,i,cols;
   inits:= [seq( [theta(0)=0, v(0)=vi, x(0)=0, y(0)=vi],
     vi=vmin..vmax, (vmax-vmin)/numv )];
   cols:= [seq(COLOR(HUE,i),i=0..1,1/numv)];
   display(
   Array([DEplot(xphug(R), [theta,v,x,y], t=0..8,
          inits,
        theta=-Pi..3*Pi, v=0..3, x=xrange, y=yrange,
       linecolor=cols, numpoints=150, obsrange=false,
        scene=[theta,v]),
    DEplot(xphug(R), [theta,v,x,y], t=0..8,
          inits,
        theta=-Pi..3*Pi, v=0..3, x=xrange, y=yrange,
       linecolor=cols, stepsize=0.1, obsrange=false,
        scene=[x,y])]))
   end:
> Plot4(1,2,3,R=1);
```




```
> Plot4(1,3,20,R=0.2,xrange=0..10,yrange=-2..3);
```




```
> xphug:=R-> [diff(theta(t),t) = (v(t)^2 - cos(theta(t))),
              diff(v(t),t)      = (-sin(theta(t))-R*v(t)^2)*v(t),
              diff(x(t),t)      = v(t)*cos(theta(t))*v(t),
              diff(y(t),t)      = v(t)*sin(theta(t))*v(t)];
```

$$xphug := R \to \left[\frac{d}{dt}\, \theta(t) = v(t)^2 - \cos(\theta(t)),\ \frac{d}{dt}\, v(t) = \left(-\sin(\theta(t)) - R\, v(t)^2\right) v(t),\ \frac{d}{dt}\, x(t) = v(t)\cos(\theta(t))\, v(t),\ \frac{d}{dt}\, y(t) = v(t)\sin(\theta(t))\, v(t) \right] \quad \textbf{(16)}$$

```
> Plot4(1,3,20,R=0.2,xrange=0..10,yrange=-2..3);
```


3
2
v
1
-2
0
2
4
6
8
θ
3
2
y
1
0
2
4
6
8
10
x
-1
-2